Labor of Fundamentals of Electronics and Power Electronics

Exercise No. 22

AUDIOFREQUENCY POWER AMPLIFIER - SIMULATIONS

22.1 Basic information

The goal of the exercise is single-stage bipolar transistor audiofrequency power amplifier model tests and digital simulations in two cases: with or without AC signals feedback. The amplifier is connected in common emitter circuit. For operating point setup voltage divider or voltage divider with emitter resistor is introduced. For simulation PSpice Student 9.1 is used.

References:

Grebene A. B.: Bipolar and MOS analogy integrated circuit design
Buckingham M. J.: Noise in electronic devices an systems
Titze U., Schenk Ch.: Semiconductors and circuits
http://en.wikipedia.org/wiki/Amplifier
http://www.ittc.ku.edu/~jstiles/412/handouts/Amplifier
Titze U., Schenk Ch.: Układy półprzewodnikowe
Baranecki A.: Laboratorium układów elektronicznych. Cz. 1
Kaźmierowski M.P., Matysik J. T.: Wprowadzenie do elektroniki i energoelektroniki
Jaczewski J., Opolski A., Stolz J.: Podstawy elektroniki i energoelektroniki
Baranowski J., Czajkowski G. Układy elektroniczne. Cz. 1

22.2 Exercise description

For tests three files are prepared: WZM1.CIR, WZM2.CIR, WZM3.CIR.

WZM1.CIR file consists amplifier model **without feedback** for AC signals, with transistor supplied by voltage divider, and with emitter resistor.

WZM2.CIR file consists amplifier model **with feedback** for AC signals, with transistor supplied by voltage divider, and with emitter resistor.

WZM3.CIR file consists amplifier model **without feedback** for AC signals, with transistor supplied by voltage divider.

In each models one transistor from another can be applied. The letters **a**, **b**, and **c** in type description mean:

BC109a transistor has current-gain factor $\beta = 100$

BC109b transistor has current-gain factor $\beta = 200$

BC109c transistor has current-gain factor $\beta = 400$

Transistor type can be chosen by corresponding letter changing in procedure name called proper transistor model (e.g. Q1 4 3 6 BC109b)

The WZM1.CIR code example is show below:

```
*****  WZM1   *****
********************
R1 2 3 150k
R3 2 4 2k
R4 6 0 200
c1 1 3 1u
c2 4 5 10u
********************
********************
Q1 4 3 6 BC109b
********************
********************
*.TRAN .1ms .2ms
********************
********************
.TRAN .5ms 2.2ms
********************
********************
********************
********************
********************
********************
********************
c3 6 0 100u
R2 3 0 30k
R5 5 0 10k
Vcc 2 0 DC 12
Vin 1 0 AC .02 sin(0 .03 1k .2m)
```

```
.model bc109a NPN(Is=10f Xti=3 Eg=1.11 Vaf=100 Bf=100 Ne=1.5 Ise=0 Ikf=0 Xtb=1.5
+  Br=1 Nc=2 Isc=0 Ikr=0 Rc=0 Cjc=2n Vjc=.75 Mjc=.3333 Fc=.5 Cje=5n Vje=.75
+  Mje=.3333 Tr=10n Tf=1n Itf=0 Vtf=0 Xtf=0)
.model bc109b NPN(Is=10f Xti=3 Eg=1.11 Vaf=100 Bf=200 Ne=1.5 Ise=0 Ikf=0 Xtb=1.5
+  Br=1 Nc=2 Isc=0 Ikr=0 Rc=0 Cjc=2n Vjc=.75 Mjc=.3333 Fc=.5 Cje=5n Vje=.75
+  Mje=.3333 Tr=10n Tf=1n Itf=0 Vtf=0 Xtf=0)
.model bc109c NPN(Is=10f Xti=3 Eg=1.11 Vaf=100 Bf=400 Ne=1.5 Ise=0 Ikf=0 Xtb=1.5
+  Br=1 Nc=2 Isc=0 Ikr=0 Rc=0 Cjc=2n Vjc=.75 Mjc=.3333 Fc=.5 Cje=5n Vje=.75
+  Mje=.3333 Tr=10n Tf=1n Itf=0 Vtf=0 Xtf=0)
.PROBE
.AC DEC 10 10 1MEG
.OPTIONS TRTOL=1 RELTOL=.000001
.END
```

22.3 Exercise program

22.3.1 Simulations' studies of an amplifier without feedback for AC signals**,** with transistor supplied by voltage divider, and with emitter resistor (WZM1.CIR file). In Fig. 22.1 a scheme of the circuit is presented.

a) For BC109b transistor and chosen supply voltage select resistor R1 value. In such a way the amplifier operates in A class. Choose **Transient analysis** type and proceed with simulations.

Because amplifier input signal is zero in time period <0; 0.2ms>. During resistor R1 selection choose **.TRAN .1ms .2ms** analysis type. The ***.TRAN .5ms 2.2ms** has to be switched-off by "*" star.

b) Choose **.TRAN .5ms 2.2ms** analysis and do simulations. Observe V(4) and V(5) voltage waveforms (block by "*****" star ***.TRAN .1ms .2ms** option). In medium frequency range check if amplifier is not saturated.

c) Set input signal amplitude, so that the amplifier would not be saturated. Choose frequency (**AC**) analysis and determine transfer characteristics: voltage gain, upper and lower cutoff frequency and bandwidth.

d) Change transistor to BC109a, check operation point position and repeat circuit simulations. Observe waveforms as described in b, and c points.

e) Change transistor to BC109c, check operations point position and repeat circuit simulations. Observe waveforms as described in b, and c points.



Fig.22.1

22.3.2 Simulations' studies of an amplifier with feedback for AC signals**,** with transistor supplied by voltage divider, and with emitter resistor (WZM2.CIR file). In Fig. 22.2 a scheme of the circuit is presented. Choose resistor R4 from 100 Ω ÷ 500 Ω range.

a) Perform a circuit simulations and waveforms observation as described in p. 22.3.1 a ÷ e. Input signal amplitude is ~0,5 V.



Fig.22.2



Fig.22.3

22.3.3 Simulations' studies of an amplifier without feedback for AC signals**,** with transistor supplied by voltage divider (WZM3.CIR file). In Fig. 22.3 a scheme of the circuit is presented.

a) Perform a circuit simulations and waveforms observation as described in p. 22.3.1 a ÷ e. Input signal amplitude is ~0,02 V.